

L3 Informatique - HLIN603
Partie III - Typage dynamique - Tout objet - Développement Agile -
méta-programmation -
avec Smalltalk.
TDs/TPs

Les notes de cours sont en : <http://www.lirmm.fr/~dony/notesCours/smalltalkOverview.s.pdf>,

Version imprimable en : <http://www.lirmm.fr/~dony/notesCours/smalltalkOverview.pdf>.

Le cours, les TDs et TP utilisent le langage Pharo, une extension d'un Smalltalk complet.

1 Téléchargez Pharo Smalltalk

Allez sur le site <http://pharo.org/>.

Téléchargement : suivez “download latest version” puis “Télécharger Pharo Launcher”. Une fois téléchargé, exécutez le programme *PharoLauncher*. Le Launcher est un programme qui permet de choisir la version des bibliothèques du langage ou même différentes applications connexes comme le cours en ligne MOOC.

Les exercices de ces TP ont été écrits pour la version “7.0 stable 64 bits” de Pharo, cela ne change rien au code mais par contre l'environnement de programmation a évolué, je vous suggère donc de choisir cette version. Cliquez à gauche sur “7.0 stable 64 bits”, puis téléchargez la (clic sur l'étoile orange). Une fois l'image téléchargée, elle apparaît dans la fenêtre de droite (voir figure , sélectionnez la et cliquez sur la flèche verte pour télécharger et installer la machine virtuelle correspondante et ouvrir l'environnement *Pharo*. Vous pouvez commencer à travailler.

Pour réouvrir l'environnement (par exemple pour le second TP) relancer le *launcher*, sélectionner l'image contenant votre travail et recliquez sur la flèche verte. Vous pouvez aussi choisir la version 8.0 plus récente dès que vous êtes à l'aise avec l'environnement.




Figure (1) – Launcher

2 La syntaxe et les bases en lisant le début du cours

Les notes de cours sont en : <http://www.lirmm.fr/~dony/notesCours/smalltalkOverview.s.pdf>,

Version imprimable en : <http://www.lirmm.fr/~dony/notesCours/smalltalkOverview.pdf>.

3 La syntaxe et les bases avec le MOOC

Vous pouvez aborder le travail en allant suivre certaines vidéos du mooc réaliser par les créateurs de Pharo (<http://mooc.pharo.org/>), en français ! Ecoutez en premier lieu l'introduction, puis allez en <http://rmod-pharo-mooc.lille.inria.fr/MOOC/WebPortal/co/pharo.html> et choisissez le chapitre qui vous intéresse, je vous recommande les sections 1, 2 et 3 pour débiter.

4 La syntaxe et les bases avec le tutoriel

L'application s'ouvre avec une fenêtre ouverte : "Welcome to Pharo xxx". Dans cette fenêtre repérer : "learn Pharo" puis "PharoTutorial go." ou "ProfStef go.". Placez le curseur derrière le point, ouvrez le menu contextuel ("command-Clic" ou "control-clic" ou "clic droit" (selon clavier et souris)), choisissez "doIt". Vous obtenez le même résultat avec le raccourci clavier "Cmd-d" ou "Control d" selon votre système. Idem pour "printIt" avec "Cmd-p" et "InspectIt" avec "Cmd-i", utiles partout et tout le temps.

Au cas où vous ne voyez pas cette fenêtre, ouvrez un *playground* (menu *Tools* \rightarrow *Playground*), entrez l'expression "ProfStef go." ¹ puis *doit* ("Cmd-d").

Le tutoriel va vous faire aller de fenêtre en fenêtre et vous présenter toute la syntaxe de base et quelques autres choses. Vous avez accès à un livre en ligne : <http://www.pharobyexample.org/>. Vous avez aussi un Mooc en ligne : <http://mooc.pharo.org>.

5 L'Environnement, premières indications

Comme indiqué en cours, l'environnement n'est pas un détail mais une part intégrante du concept, permettant dans la vraie vie de programmer "in the large" vite et bien.

- Menu *World* : clic gauche sur fond d'écran. Les items essentiels dans un premier temps sont *Tools-System Browser*, *Tools-Playground* et *Tools-Transcript* si vous voulez afficher des messages ; par exemple (Transcript show: 'Hello World'; cr.).
- Chaque sous-genêtre de chaque outil possède un menu contextuel, "Cmd-clic" ou "Ctrl-clic" ou clic bouton droit souris.
- Sauvegarde de vos travaux : *menu Pharo-save*. Nous ferons plus subtil ultérieurement.
- Ouvrez un **System Browser**, dans le menu contextuel de sa fenêtre en haut à gauche, faites "Find Class" et cherchez la classe *OrderedCollection*. Une fois sélectionnée, regardez la liste de ses méthodes et le classement en catégories. Les catégories sont un concept de l'environnement ; elles n'ont pas d'incidence sur l'exécution des programmes.
- Ouvrez un *Playground*, c'est comme un tableau de travail, entrez des expressions, choisissez "doIt", "print It" ou "inspectIt" pour exécuter, exécuter et afficher le résultat ou exécuter et inspecter le résultat. Ceci vaut pour toute expression. Toute instruction est une expression.

```
1 t := Array new: 2.  
2 t at: 1 put: #quelquechose.  
3 t at: 1  
  
5 c := OrderedCollection new: 4
```

1. Notez que c'est un envoi de message qui demande à ProfStef de commencer son cours.

```

6 1 to: 20 do: [:i | c add: i]
7  "even dit si un nombre est pair"
8  c count: [:each | each even]
9  "aller vous ballader sur la classe Collection pour regarder les
10 itérateurs disponibles"

```

6 Classes, instances, méthodes d'instance

Pour se familiariser, création d'une classe simple. Ouvrez un *System Browser*, dans le menu contextuel de sa fenêtre haut-gauche, faites "Add Package", donnez lui un nom, par exemple *HLIN603*.

1. A définir la classe *Pile* implantée par composition avec un tableau (*Array*, ce sera ainsi dans le corrigé) ou une *OrderedCollection* qui va bien aussi.

— Sélectionnez votre package et pour créer la classe, cliquez dans la seconde fenêtre du browser, renseignez le *template*, puis "accept". Ensuite dans le *playground* essayez *Pile new* "inspectIt".

```

1 Object subclass: #Pile
2   instanceVariableNames: 'contenu index capacite'
3   classVariableNames: 'tailleDefaut'
4   category: 'HLIN603'

```

- Définissez la méthode *initialize: taille*, équivalent d'un constructeur à 1 paramètre, qui initialise les 3 attributs (dits "variables d'instance"). Essayez ensuite *: Pile new initialize: 5*.

```

1 initialize: taille
2   "la pile est vide quand index = 0"
3   index := 0.
4   "la pile est pleine quand index = capacite"
5   capacite := taille.
6   "le contenu est stocké dans un tableau"
7   contenu := Array new: capacite.
8   "pour les tests, enlever le commentaire quand isEmpty est écrite"
9   "self assert: (self isEmpty)."

```

- Ecrivez les méthodes *: isEmpty*, *isFull*, *push: unObjet*, *pop*, *top*. Testez les dans le *playground*.
- Pour la rendre compatible avec le *printIt* de l'environnement, définissez la méthode suivante sur la classe. C'est l'équivalent du *toString()* de Java. L'opérateur de concaténation des chaînes est " , " (par exemple 'ab' , 'cd').

```

1 printOn: aStream
2   aStream nextPutAll: 'une Pile, de taille: '.
3   capacite printOn: aStream.
4   aStream nextPutAll: ' contenant: '.
5   index printOn: aStream.
6   aStream nextPutAll: ' objets : ('.
7   contenu do: [:each | each printOn: aStream. aStream space ].
8   aStream nextPut: $).
9   aStream nextPut: $..

```

— Signalez les exeptions, en première approche, vous écrirez : `self error: 'pile vide'..`

2. Apprenez à utiliser le débogueur. Insérer l'expression `self halt.` au début de la méthode `push.`. Après lancement, l'exécution s'arrête à ce point, choisissez "debug" dans le menu proposé. Vous voyez la pile d'exécution. Vous pouvez exécuter le programme en pas à pas (les items de menu importants sont "into" et "over" pour entrer, ou pas, dans le détail de l'évaluation de l'expression courante. Le débogueur est aussi un éditeur permettant le remplacement "à chaud". Le debugger d'Eclipse a été construit sur le modèle de celui-ci.

7 Composition et méthodes (de type *keyword*) à plusieurs paramètres

Réaliser une classe distributeur de Bonbons (type foire foraine) à n colonnes modélisée comme un tableau de n piles (chaque colonne est représentée par une pile).

- Création d'un distributeur de 2 colonnes : `D:= (Distributeur new) colonnes: 2 taille: 5.`
- Créer une classe `Bonbon` et deux sous-classes `Carambar` et `Malabar` (par exemple).
- Ecrire la méthode `remplir:avec:` telle que `D remplir: 1 avec: Carambar.`, remplisse la colonne 1 avec des instances de la classe `Carambar`; ceci suppose l'instantiation de la classe passée comme second argument de la méthode `remplir:avec:`.
- Ecrire la méthode `donner:` telle que `D donner: 1.` rende le premier élément de la colonne 1 s'il en reste (donc *aCarambar*, sinon *#YenAPlus*).

8 Jeux de Test

Typage dynamique et développement Agile ou Extrême vont de pair avec la réalisation systématique de test après chaque modification. La définition des tests est souvent préalable à l'écriture du code; les tests faisant partie de la spécification. Pharo intègre une solution rationnelle pour organiser des jeux de tests dans l'espace (couverture du code) et le temps (rejouer les tests après une modification du code).

1. Appliquez le tutoriel ci-dessous au cas de la pile en créant une classe `TestPile`. Il faut pour cela créer, dans le même package que l'application une sous-classe de `TestCase`, comme indiqué en : <http://pharo.gforge.inria.fr/PBE1/PBE1ch8.html>.
2. Si vos tests ne passent pas (couleur rouge), le bon outil pour déboguer est le *TestRunner* (menu principal).

9 Méthodes de classes (Take a first walk on the wild side)

Les méthodes de classe s'exécutent en envoyant des messages aux classes (ainsi considérées comme des objets. Par exemple `Date today`. Pour observer ou définir des méthodes de classes, il faut cliquer sur le bouton "class" du browser.

- Avant de passer côté *class*, ajouter une variable de classe `tailleDefaut` à la classe `Pile` (cela se fait côté instance - demandez vous pourquoi.).
- Définir une **méthode de classe** `initialize` qui fixe à 5 la taille par défaut des piles, valeur 5 stockée dans la variable de classe `tailleDefaut`. Vous devez exécuter cette méthode pour que la variable de classe soit initialisée. De par son nom (`initialize`) cette méthode est reconnue par le browser (voir flèche verte en face du nom). Si vous décidez de la nommer autrement, vous aurez à lancer cette exécution par : (`Pile initialize`).

- Définir sur `Pile` la méthode de classe `new`: pour qu'elle appelle la méthode **d'instance** `initialize`: définie en section 6.
- Définir sur `Pile` la méthode de classe `new` pour qu'elle appelle la méthode **de classe** `new`: définie juste avant.
- Définir une méthode de classe `example` réalisant un exemple de programme utilisant une pile.

10 Héritage - Redéfinitions

Pour expérimenter l'héritage, on va reprendre l'exercice des comptes bancaires fait en C++ et Ocaml.

On traite dans cette section des classes `Account`, `InterestAccount` et `SecureAccount`.

On traitera de la classe `Bank` dont les instances possèdent une collection de comptes en section 11.

1. Définir `Account` puis ses méthodes `initialize`, `deposit:`, `withdraw:`, et `get`.

Je vous donne la méthode `printOn`: (notez le joli `self class name`, pour la réutilisation).

```
1 printOn: aStream
2     aStream nextPutAll: 'un ', self class name, ' de solde : ' .
3     balance printOn: aStream.
```

2. **Redéfinition 1.**

Ouvrez un second browser, visualisez la méthode `=` de la classe `Object`. Repérez le petit triangle bleu à la gauche du nom qui permet d'identifier puis de visualiser toutes ses redéfinitions.

Redéfinissez la méthode `=` sur `Account` (notez que c'est une méthode binaire, selon la définition du cours *Ocaml*). Notez que vous avez un problème pour accéder au `solde` du compte reçu en paramètre, car les attributs sont *protected* en *Smalltalk*; comment est-ce solutionné en Java, C++, Ocaml? Ici il faut utiliser la ruse ou l'introspection (voir la méthode `instVarAt:` de `Object`). Je suggère la ruse.

3. **Redéfinition 2.**

Redéfinissez `withdraw:` sur `SecureAccount` afin qu'il soit impossible de retirer d'un *secureAccount* si le solde est insuffisant.

4. **Redéfinition 3.**

Redéfinissez `deposit:` sur `InterestAccount` avec le code ci-dessous, et définissez `depositInterest:` qui ajoute sur le compte receveur 5% de la somme déposée `n`.

```
1 "méthode d'instance de la classe InterestAccount"
2 deposit: n
3     super deposit: n.
4     self depositInterest: n.
```

11 Collection hétéroclite (collection de comptes)

Où l'on voit que l'on peut faire du polymorphisme paramétrique en typage dynamique. Il n'y a aucune restriction et aucun contrôle de type. La seule limite dans l'usage que l'on fait des éléments qui sont stockés dans la collection; ils doivent tous être capables de répondre au message qui leur sera envoyé lors d'une itération.

1. Définissez la classe **Bank** qui définit les objets possédant un ensemble de comptes stockés dans un attribut **accounts** initialisé avec une *OrderedCollection*.
2. Définissez les méthodes.
 - **add:**, ajoute un compte à une banque ;
 - **balance**, calcule la somme des soldes des comptes du receveur ;
 - **deposit: n**, depose la somme n sur chacun des comptes du receveur.
3. Une bonne habitude à prendre : écrivez la méthode **printOn:**.
4. Définissez et exécutez la méthode de classe **example** ci-dessous.

```

1 example
2     "une méthode de classe de la classe Bank"
3     "Bank example" "Selectionnez le commentaire precedent puis doIt"
4     | b |
5     b := self new initialize.
6     b add: (SecureAccount new: 200). "pourquoi les parenthèses ?"
7     b add: InterestAccount new.
8     b add: (SecureAccount new: 150).
9     b deposit: 100.
10    ^b

```

5. En l'exécutant, vérifiez que la liaison dynamique fonctionne correctement dans le cas d'envoi de messages à des éléments d'une collection hétérogène, donc que la méthode **depositInterest:** est bien invoquée pour chaque type de compte et qu'elle ajoute des intérêts sur le bon compte.
6. Pour voir plus d'itérateurs, ajoutez à la classe **Bank** les méthodes : **fees**, enlève 5% de frais à tous les comptes ; **min**, rendant le compte ayant le plus faible solde.
Pour cela examinez la collection des itérateurs (protocole *enumetating*) sur la classe **Collection**.

12 Contrôle de types à l'exécution

Typage dynamique ne signifie pas absence de types. Les types sont présents à l'exécution où tout objet possède un type défini par sa classe (tout est objet et tout objet est instance d'une classe). Il est donc possible de programmer des contrôles si on le souhaite².

1. Exemple : écrire une classe **PileTypée**, sous-classe de la classe **Pile** qui permet d'imposer le type des éléments empilés. En pratique il faut pour cela utiliser un attribut pour stocker la classe des éléments à empiler, une méthode d'instance (et une de classe) pour l'initialiser, et une redéfinition de la méthode **push:**.

Utilisation :

```

1 p := PileTypee de: Bonbon.
2 p push: Carambar new.
3 p push: 22 "--> exception, cette pile ne peut contenir que des Carambar"

```

2. (travail post TP) Etendre l'exercice précédent au cas du distributeur en le reprogrammant pour en faire une version non contrainte et une version contrainte utilisant des piles typées.

2. Il est à noter que ce n'est pas dans la pratique avec de tels langages où la qualité des programmes découle des tests systématiques et du fait qu'il n'est pas si fréquent de placer par erreur un chien dans une collection de chats. L'intérêt de l'exercice est intéressant de voir comment manipuler les types dans un langage tout objet.

13 Tout est objet : nil comme la liste ou l'arbre vide

`nil` est la valeur par défaut contenue dans tout mot mémoire géré par la machine virtuelle *Smalltalk*. Toute variable ou attribut ou case de tableau non initialisée contient `nil`. En *Smalltalk*, `nil` est aussi un objet, l'unique instance de la classe `UndefinedObject` (dont le nom me semble faire peu de sens puisque `nil` est parfaitement défini mais c'est un point de vue personnel). On peut donc envoyer des messages à `nil` qui correspondront à des méthodes définies sur `UndefinedObject`.

Par ailleurs, cet exercice est l'occasion de bien noter que l'envoi de message réalise un test de type implicite, à comparer à un test de type explicite réalisé dans un *case-switch*).

1. En utilisant cette information, programmez la classe `ArbreBinaireDeRecherche` (ou `ABR`), selon l'énoncé du TD2 Ocaml - question 4, en définissant toutes les méthodes relatives aux arbres vides sur la classe `UndefinedObject`. (pas d'obligation à traiter le cas du *remove* si vous n'avez pas de temps, il relève plus du cours d'algorithmique scripto sensu).

NB : **Environnement**. Si `HLIN603` est le nom de votre package de travail en TP, définissez la classe `ABR` dans le package nommé *HLIN603-ABR* et définissez les méthodes sur la classe `UndefinedObject` dans le protocole (ou catégorie) de méthodes nommé **HLIN603-ABR*. Ainsi vous pourrez tout visualiser au même endroit (le package `HLIN603`) dans le browser.

2. Définissez un itérateur `do` : pour les arbres binaires de recherche.

14 Méta-programmation et IDM

La méta-programmation est la programmation des entités de niveau méta³ opérée avec le langage standard. Elle suppose que les entités de niveau méta soient accessibles aux programmes standards, donc que le langage soit réflexif. On appelle méta-objet un objet représentant une entité du niveau méta.

14.1 Inspections

Inspectez la classe `Pile`, puis son dictionnaire des méthodes, puis sa méthode `push`..

Inspectez le résultat de l'expression : `Pile compiledMethodAt: #push`..

Trouvez la classe sur laquelle est définie la méthode `compiledMethodAt`..

14.2

14.3 Un programme qui fabrique un programme

But de l'exercice : créer par programme la classe `Cpt` représentant les compteurs, puis l'instancier puis envoyer un message à une instance. Les compteurs sont des objets détenant une valeur qu'ils peuvent incrémenter ou décrémenter.

Pour cela, créez une classe `CreateCpt`. Sur cette classe créez la méthode de classe `do` suivante, testez la et finissez la.

3. par exemple, une classe, le dictionnaire des méthodes d'une classe, l'arbre de syntaxe d'une méthode, la pile d'exécution de la machine virtuelle, le parseur, le compilateur, le browser, ... , toutes les entités qui permettent la fabrication et l'exécution des programmes.

```

2 do
3   "fabriquer une classe et ses méthodes par programme"
4   "détruisez la classe Cpt si elle existe avant de relancer l'exécution"
5   "CreateCpt do"
6
7   | newClass unCpt initializeMethod |
8
9   "créer la classe Cpt"
10  Object
11    subclass: #Cpt
12    instanceVariableNames: 'val'
13    classVariableNames: ''
14    package: 'HLIN603'.
15
16  "référencer la classe Cpt dans une variable"
17  newClass := Smalltalk classNamed: #Cpt.
18
19  "créer une instance"
20  unCpt := newClass new.
21
22  "fabriquer le code et compiler la méthode initialize de Cpt"
23  initializeMethod := OpalCompiler new
24    source: 'initialize\ ^val := 0' withCRs;
25    class: newClass;
26    compile.
27
28  "ajouter la méthode à la classe"
29  newClass addSelector: #initialize withMethod: initializeMethod.
30
31  "exécuter la méthode et vérifier la valeur rendue"
32  self assert: (unCpt initialize == 0).
33
34  "à vous de continuer"
35  "commencer par programmer en début de méthode : détruire la classe Cpt si elle existe"

```

14.4 Accéder à la pile d'Exécution de la machine virtuelle

Implantez sur la classe `Symbol`, les méthodes `catch` et `returnToCatchwith:` suivantes qui accèdent à la pile d'exécution via la pseudo-variable *thisContext*.

```

1 !Symbol methodsFor: 'catch-throw'!
2
3 catch: aBlock
4   "execute aBlock with a throw possibility"
5   aBlock value.
6
7 returnToCatchWith: aValue
8   | catchMethod currentContext |
9   currentContext := thisContext.
10  catchMethod := Symbol compiledMethodAt: #catch:.
11  [currentContext method == catchMethod and: [currentContext receiver == self]]

```



```

12         whileFalse: [currentContext := currentContext sender].
13         currentContext return: aValue.
14         ^aValue

```

Listing (1) – version Pharo-6

Exemple d'utilisation dans le *playground* :

```

1 ^#Essai catch: [
2     Transcript show: 'a';cr.
3     Transcript show: 'b';cr.
4     Transcript show: 'c';cr.
5     #Essai returnToCatchWith: 22.
6     Transcript show: 'd';cr.
7     33]

```

Listing (2) – should display a b c (not d) in the Transcript and return 22 (not 33)

Vous pouvez ensuite lire la méthode `signal` (équivalent de `throw`) de la classe `Exception`.

14.5 Réflexion de comportement, les méta-objets pour modifier l'exécution

```

1 "la méthode ast rend l'arbre de syntaxe abstraite du receveur"
2 node := (Pile compiledMethodAt: #push:) ast.

4 "Il est possible d'associer un méta-objet à tout noeud de l'arbre de syntaxe"
5 link := MetaLink new metaObject: (Object new); selector: #intercepte.
6 node link: link.

8 (Pile new: 3) push: 33.

```

Listing (3) – “Reflection in Pharo : Beyond Smalltalk”, Marcus Denker

14.6 Création d'une nouvelle méta-classe

Modifiez la classe `Pile` pour qu'elle mémorise la liste de ses instances dans une collection stockée dans une variable d'instance de la métaclasse, et dotée d'un accesseur sur cette collection.

15 Tester les fermetures

1. Testez les exemples du cours relatifs aux blocks.
2. Ecrivez sur une classe `Counter` une **méthode de classe** `create` :

```

1 create
2     | x |
3     x := 0.
4     ^ [ x := x + 1 ]

```

3. Appelez deux fois la méthode et stockez les valeurs rendues dans 2 variables. Exécutez plusieurs fois les blocks contenus dans ces deux variables.

```
1 c1 := Counter create.  
2 c1 value.  
3 c1 value.  
4 c2 := Counter create.  
5 c2 value.
```

4. Inspectez ces deux variables en faisant le lien avec la définition de la classe `BlockClosure`.

16 Héritage et Traits

Lisez la section 3.11 du cours.

On souhaite pouvoir trier la collection de comptes d'une banque par la balance de ses comptes, du plus bas au plus élevé. On définit la méthode `sort` sur `Banque` avec comme code : `accounts sort`.

Sachant que `accounts` est une `OrderedCollection`, que `OrderedCollection` hérite de `SequenceableCollection`, que `SequenceableCollection` utilise le trait `TSortable` dont le code est ci-dessous, déduisez en ce qu'il faut faire sur la classe `Account` pour obtenir le résultat demandé.

```
1 Trait named: #TSortable  
2   uses: {}  
3   category: 'Collections-Abstract'  
  
5 !TSortable methodsFor: 'sorting' !  
6 sort  
7   "Sort this collection into ascending order using the '<=' method."  
8   self sort: [:a :b | a <= b]
```