Université Montpellier-II UFR des Sciences - Département Informatique

Objets Avancés

Vue générale de Smalltalk : Tout Objet, Typage Dynamique, Réflexivité, Métaprogrammation, IDM, Environnement de Génie Logiciel

> Notes de cours - 2007-2019 Christophe Dony

1 Introduction

1.1 Idées clé

- 1. Simplicité (4 idées), typage dynamique \rightarrow réutilisation, agilité, prototypage rapide
- 2. Environnement intégré de génie logiciel \rightarrow efficacité, agilité, prototypage rapide, extreme programming




Figure (1) - Environnement Pharo-6.1

3. Langage réflexif \rightarrow Systèmes auto-adaptables, Modèles à l'exécution

1.2 Liens historiques et pratiques

- Alan Kay (https://fr.wikipedia.org/wiki/Alan_Kay).
- Histoire, l'idée de base : https://fr.wikipedia.org/wiki/Dynabook

- Histoire, les livres : http://www.world.st/learn/books
- La machine virtuelle pour le tout objet réflexif : CoG de Eliot Miranda (http://www.mirandabanda.org/cogblog/about-cog/)
- Le centre d'information Européen : http://www.esug.org/
- Le langage utilisé utilisé dans ces notes : http://pharo.org/ (Pharo is a clean, innovative, open-source Smalltalk-inspired environment)
- Un cours interactif en ligne: http://mooc.pharo.org,
- http://stephane.ducasse.free.fr/FreeBooks.html : catalogue de livres sur Smalltalk dont certains en ligne
- www.squeak.org: Squeak est un Smalltalk gratuit réalisé par une équipe dirigée par Alan Kay
- les versions industrielles : http://www.cincomsmalltalk.com/main/ : Cincom, Smalltalk industriel (ObjectStudio Visualworks)
- http://www.threeriversinstitute.org : site de Kent Beck, fondateur du "Agile development" avec Smalltalk.
- POURQUOI?:

```
https://en.wikipedia.org/wiki/Extreme_programming.
http://stackoverflow.com/questions/1821266/what-is-so-special-about-smalltalk
https://blog.appacademy.io/pharo-future-software-development/
```

1.3 Le langage

- Objet : héritier de Simula (abstraction de données, polymorphisme d'inclusion)
- héritier de *Lisp*
 - typage dynamique (pas de problèmes de redéfinitions covariantes, peu de contrôles statique),
 - passage par référence généralisé,
 - allocation et récupéraration dynamiques et automatiques de la mémoire,
 - style applicatif (toute instruction est une expression),
 - compilation en instructions d'une machine virtuelle,
 - réflexif : programmes = données,
 - méta-programmable.

Le langage en quatre idées

1. Toute entité est un objet.

Un objet est une entité individuelle, repérée par une adresse unique, possédant un ensemble de champs (autant que d'attributs sur la classe), connus par leurs noms et contenant une valeur qui peut changer (mutable).

- 2. Tout objet est instance d'une classe, qui définit sa structure (définie par un ensemble attributs privés) et ses comportements (définis par un ensemble de méthodes).
- 3. Tout comportement d'un objet (méthode) est activé par un envoi de message (avec liaison dynamique).
- 4. Toute classe (sauf ProtoObject) est définie comme une spécialisation d'une autre. La relation de spécialisation définit un arbre d'héritage.

1.4 Interprétation, Compilation, machine virtuelle

Principe d'exécution : compilation des instructions en byteCodes ou instructions d'une machine virtuelle dédiée à la programmation par objets, puis interprétation de ces instructions.

Machine virtuelle Smalltalk : ensemble d'instructions dédiées à la programmation par objets et interpréteur associé. Java a repris ce schéma d'exécution. Voir par exemple COG (http://www.mirandabanda.org/cogblog/about-cog/).

Exemple de méthode et de bytecode généré

Listing (1) - définition de la méthode fact sur la classe PositiveInteger

```
normal CompiledMethod numArgs=0 numTemps=0 frameSize=12
    literals: (#fact )
    1 < 44 > push self
    2 < 49 > push 0
    3 < A6 > send =
    4 < EC 07 > jump true 13
    6<\!\!44\!\!>\mathrm{push}\;\mathrm{self}
    7 < 44 > push self
    8 < 4A> push 1
    9 < \text{A1} > \text{send}
10
    10 < 70 > send fact
11
    11 < A8 > send *
12
13
    12 < 66 > pop
    13 < 60 > push self; return
```

1.5 Introduction pratique

Classes et instances

```
Object subclass: #Compteur
2
       instanceVariableNames: 'valeur'
3
4
       classVariableNames: ''
       poolDictionaries: ''
5
       category: 'ExosPharo-PetitsExercices'
6
   initialize
8
       valeur := 0.
9
10
   get
        ^valeur
11
12
   incr
       valeur := valeur + 1.
13
   decr
14
       valeur := valeur - 1.
15
```

Listing (2) - classe Compteur, définition de classe et de méthodes d'instance

```
new
2
        "redéfinition de la méthode new pour appeler automatiquement une méthode d'initialization"
3
        <sup>^</sup>super new initialize
4
    example
6
        | c |
7
        c := Compteur new.
8
        c incr.
9
         ^C
10
```

Listing (3) – définition de méthodes de classe

1.6 La classe Compteur dans l'environnement




Figure (2) - Définition de la classe Compteur - Pharo Env.

1.7 L'exception universelle : doesNotUnderstand

3.4 factorial

 \rightarrow Instance of SmallFloat64 did not undestrand factorial

Conduite à tenir :

- lire le message
- si non compréhension, ouvrir le debbugger (one click), corriger, relancer

1.8 Tout objet et réflexivité

```
1 class "-> SmallInteger"
2 1 class class "-> SmallInteger class"
3 1 class class class "-> Metaclass"
```

```
1 class class class class "->Metaclass class"

1 class class class class class "->Metaclass"

5 factorial "->120"

m := Integer compiledMethodAt: #factorial.

m class "->CompiledMethod"

m name "->'Integer>>#factorial"

m valueWithReceiver: 5 arguments: #() "->120"

factorial

^self = 0

ifTrue: [thisContext inspect.

1]

ifFalse: [self * (self - 1) fact]
```

2 Syntaxe

Listing (4) - from: http://wiki.c2.com/?SmalltalkSyntaxInaPostcard

- http://www.chimu.com/publications/JavaSmalltalkSyntax.html: Une comparaison des syntaxes Java et Smalltalk.
- http://www.csci.csusb.edu/dick/samples/smalltalk.syntax.html : Une présentation concise de la syntaxe de Smalltalk.

2.1 Constantes littérales

"Constantes" car leur valeur ne peut être modifiée et "littérales" car elles peuvent être entrées littéralement dans le texte des programmes.

```
nombres: 3, 3.45, -3,
caractères: $a, $M, $$
chaines: 'abc', 'the smalltalk system'
```

- symboles: #bill, #a22
- booléens : true, false
- tableaux de constantes littérales (construit à l'analyse syntaxique) : #(3 3.45 -3 aM 'abc' #bill true #(am stram gram))
- commentaires : "ce commentaire sera ignoré par l'analyseur syntaxique"

2.2 Identificateur

- Suite de chiffres et lettres commençant par une lettre.
- Manipulables dans le texte des programmes via leur symbole éponyme.

```
Integer methodDictionary at: \#factorial
```

— Non typés dans le texte des programmes (typage dynamique).

2.3 Affectation

```
i := 3
```

Listing (5) - à la Algol, ou à la Pascal

2.4 Séquence d'Instructions

Il y a une séquence implicite associée à tout bloc (corps de méthode par exemple). Dans un bloc, Les instructions sont séparées par un "." (et pas un ";").

```
 \begin{array}{l} \mathtt{uneMethode} \\ \mathtt{i} := 1. \\ \mathtt{j} := 2. \end{array}
```

2.5 Syntaxe spécifique pour création de tableaux à l'exécution

```
\begin{array}{l} {\tt a} := 3. \\ \{ \ {\tt a} \ . \ {\tt a}{+}1 \ . \ {\tt a}{+}2 \} \ "--> \# (3 \ 4 \ 5) \end{array}
```

2.6 Retour à l'appelant

```
^33 "(return)"
```

Toute invocation de méthode rend une valeur, self par défaut, le type void n'existe pas.

2.7 Envoi de message

Définitions:

- Conceptuelle : demande à un objet d'exécuter un de ses comportements implanté sous la forme d'une méthode,
- Pratique : appel de méthode avec sélection selon le type dynamique du premier argument qualifié de receveur et syntaxiquement distingué.

Trois sortes de messages (distinction syntaxique):

```
    unaires (à un seul argument : le receveur)
    class
    fact
```

2. binaires (à deux arguments, syntaxe pratique pour les opérations arithmétiques en infixé)

```
1 + 2
```

3. "keywords" a $n(n \ge 2)$ arguments dont le receveur.

Leur forme syntaxique (originale) permet de représenter un envoi de message comme une "petite conversation" (small talk) entre un objet et un autre.

```
1 log: 10
   anArray at: 2 put: 3
   monCalendrier enregistre: unEvenement le: unJour à: uneheure
   Les familiers de Java traduiront en :
   monCalendrier.enregistre(unEvenement, unJour, uneHeure)
   Noms des méthodes utilisées dans ces exemples :
   log:,
   at:put:,
   enregistre:le:à:.
 - Précédence :
   unaire > binaire > keyword,
   exemple: 1 + 5 fact égale 121 et pas 720.
— Associativité : A précédence égale les messages sont composés de gauche à droite.
   exemple:
   2 + 3 * 5 = 25
   2 + (3 * 5) = 17
— Cascade de messages :
   r s1; s2; s3. est équivalent à : r s1. r s2. r s3.
```

2.8 Structures de contrôle

Il n'existe aucune forme syntaxique particulière pour les structures de contrôle ; le contrôle se réalise par envois de messages.

Le propre d'une structure de contrôle est d'avoir une politique non systématique d'évaluation de ses arguments. Implanter des structures de contrôle comme des méthodes standard nécessite donc que les arguments de ces méthodes ne soient pas évalués systématiquement au moment de l'appel. La solution Smalltalk à ce problème consiste à utiliser des *blocks* (fermetures lexicales), soit en position de receveur, soit en position d'argument.

Une fermeture lexicale est une fonction anonyme (lambda-expression) capturant son environnement lexical de définition (toute variable libre y est interprétée relativement à l'environnement dans lequel la fermeture a été définie).

La syntaxe de définition d'une fermeture est [<parametre>* | <instruction>*]

2.8.1 Exemples

Conditionnelle

```
(\texttt{aNumber} \setminus \setminus 2) = 0 \; \texttt{ifTrue} \colon [\texttt{parity} := 0] \; \texttt{ifFalse} \colon [\texttt{parity} := 1]
```

Boucle "For"

La boucle for utilise un block à un paramètre qui tient lieu de compteur de boucle

```
1 to: 20 do: [:i | Transcript show: i printSting].
```

Itérateur généralisé

```
untableau := \#(3\ 5\ 7\ 9).
untableau do: [:each | Transcript show: each; cr]
```

Formes de gestion d'exceptions

```
untableau findKey: x ifAbsent: [0]
```

3 Pratique basique du langage

3.1 Classes et Méthodes

Toute classe est créée comme sous-classe d'une autre classe.

Variables d'instance, variables de classes (équivalent des "statiques" de C++ et Java), variables de pool (pas d'équivalent en Java (à l'heure où j'écris ces lignes)).




Figure (3) - Définition de la classe TimeSpan - Pharo env.

Méthodes d'instance




Figure (4) - Méthode d'instance "=" de la classe TimeSpan

Méthodes de classe

Méthodes invocables par envoi de message aux classes ¹

```
Date new. Date today.
```

^{1.} Les méthodes de classe sont en fait des méthodes d'instance définies sur les métaclasses, voir la section 4.3. En première approche, elles ressemblent aux *static* en C++ puis Java, mais elles sont en fait des méthodes standard définies sur les classes des classes (tout est objet), utilisables de façon standard.




Figure (5) – Méthode de classe "today" de la classe Date (En fait une méthode d'instance de la (méta)-classe "Date class")

Différentes sortes de variables accessibles au sein des méthodes

Les identificateurs accessibles au sein d'une méthode M :

- les paramètres de M,
- les variables temporaires de M,
- les attributs (variables d'instance) définis sur la classe du receveur,
- les identificateurs prédéfinis self, super, true, false, nil.
- les variables de classe de la classe C de O (celle ou est définie M)
- les variables partagées entre la classe C et d'autres classes (voir PoolDictionary),
- les variables globales du système, dont celles référençant toutes les classes.

3.2 Envoi de message

Demande à un objet d'exécuter un de ses services.

Procédé, réalisé à l'exécution, menant a l'exécution d'un méthode de même nom que le selecteur.

Consiste en la recherche de la méthode dans la classe du receveur du message puis en cas d'échec dans ses superclasses.

Diverses techniques d'optimisation de l'envoi de messages sont implantées dans les diverses machines virtuelles dont la première est la technique de cache (voir le green book).

3.3 Instantiation et initialisation des objets

- On instancie une classe en lui envoyant le message new.
- new est une méthode définie sur une superclasse commune à toutes les métaclasses (cf. métaclasses). new réalise l'allocation mémoire et rend la référence sur le nouvel objet.
- Exemple : Date new, d := Date new.

Il n'y a pas de constructeurs.

L'initialisation s'effectue via des méthodes standard, souvent invoquées dans des redéfinitions de la méthode de classe new.

Exemples : les méthodes new et x:y: de la classe Point.

3.4 Exemple

```
Object subclass: #Compteur
1
       instanceVariableNames: 'valeur'
2
       classVariableNames: ''
3
4
       poolDictionaries: ''
       category: 'Compteur-MVC'
5
   !Compteur methodsFor: 'acces lecture/ecriture'!
7
   nombreUnites
        ^valeur
9
   nombreUnites: n
11
       valeur := n.
12
   !Compteur methodsFor: 'imprimer'!
   printOn: aStream
16
17
       aStream nextPut: $0.
       self nombreUnites printOn: aStream
18
   !Compteur methodsFor: 'operations'!
   decr
3
           self nombreUnites: valeur -1
4
6
   incr
7
           self nombreUnites: valeur +1
9
   raz
       self nombreUnites: 0
10
11
   Compteur class
13
       instanceVariableNames: ''
14
   !Compteur class methodsFor: 'creation'!
16
17
   new
        super new raz
```

3.5 Classe abstraite

Une classe ne peut être déclarée abstraite mais elle peut être rendue abstraite par le programmeur en redéfinissant la méthode new pour signaler une exception.

Ceci pose néanmoins de sérieux problèmes pour ses futures sous-classes concrètes. (utilisation obligatoire de basicNew). Ceci est une limite de la solution Smalltalk pour les métaclasses (voir section 4.3).

3.6 Methode abstraite

Une méthode ne peut être déclarée abstraite mais elle peut être rendue abstraite en la définissant de la façon suivante :

```
method
self subclassResponsibility
```

3.7 Polymorphisme d'inclusion et Héritage

- Créer une sous-classe SC d'une classe C, revient à définir un sous-type, au sens du polymorphisme d'inclusion², du type définit par C. Ainsi, SC hérite de C et toute instance de SC possède les caractéristiques définies par C.
- L'Héritage est simple.
- La redéfinition des méthodes est basée sur leur nom. Toute définition sur une sous-classe d'une méthode de même nom existant sur une superclasse est une redéfinition.
- Typage dynamique:
 - + pas de surcharge,
 - + pas de problème de redéfinition covariante, pas de downcast,
 - - aucun contrôle statique de correction des types.
- avantages et défauts à corréler à la pratique de tests systématiques et élaborés? (Plus globalement voir *Agile Development*, ou la classe TestCase en *Pharo*).

3.8 Schémas de spécialisation

- Masquage : spécialisation d'une méthode sur une sous-classe.
- Masquage partiel : utilisation de la pseudo-variable super.
- Paramétrage par spécialisation (pseudo-variable self), ou par composition, classiques (voir cours "Réutilisation/Fra

3.9 Un exemple de hiérarchie : Les collections

```
ProtoObject #()
1
        Object #()
2
            Collection #()
                Bag #('contents')
5
                    IdentityBag \#()
6
                CharacterSet #('map')
8
                SequenceableCollection #()
                     ArrayedCollection #()
9
10
                         Array #()
                             ActionSequence \#()
11
                             DependentsArray #()
                             WeakActionSequence #()
13
                             WeakArray \#()
14
                        Array2D #('width' 'contents')
15
16
                        B3DPrimitiveVertexArray #()
                    Bitmap #()
Heap #('array' 'tally' 'sortBlock')
17
18
                     Interval #('start' 'stop' 'step')
19
                        TextLineInterval #('internalSpaces' 'paddingWidth' 'lineHeight' 'baseline')
20
                     LinkedList #('firstLink' 'lastLink')
```

^{2.} L. Cardelli and P. Wegner, "On Understanding Types, Data Abstraction and Polymorphism", ACM Computing Surveys, vol. 17, No 4, 1985, pages 471,522.

```
Semaphore #('excessSignals')
22
                       MappedCollection #('domain' 'map')
OrderedCollection #('array' 'firstIndex' 'lastIndex')
23
24
                            GraphicSymbol #()
25
                            SortedCollection #('sortBlock')
26
                            UrlArgumentList #()
27
28
                       SourceFileArray \#()
                            StandardSourceFileArray #('files')
29
30
                  Set #('tally' 'array')
                       Dictionary \#()
31
                           HtmlAttributes #()
32
                            IdentityDictionary #()
                                34
35
                                         SmalltalkEnvironment #()
36
37
                            LiteralDictionary \#()
                            MethodDictionary \#()
38
                           \label{eq:pluggableDictionary} \begin{array}{l} \texttt{PluggableDictionary} \ \#(\text{'hashBlock' 'equalBlock'}) \\ \texttt{WeakKeyDictionary} \ \#() \end{array}
39
40
                                ExternalFormRegistry #('lockFlag')
41
                                {\tt WeakIdentityKeyDictionary}~\#()
42
                            WeakValueDictionary \#()
43
                       IdentitySet #()
44
                       PluggableSet #('hashBlock' 'equalBlock')
45
                       WeakSet #('flag')
                  SkipList #('sortBlock' 'pointers' 'numElements' 'level' 'splice')
IdentitySkipList #()
47
48
                  WeakRegistry #('valueDictionary' 'accessLock')
49
```

3.10 Héritage et description différentielle




Figure (6) – Hiérarchie d'héritage avec ajouts de fonctionnalités. Pour plus de détails sur la description différentielle, voir notes de cours "Réutilisation/Frameworks".

3.11 Traits

Les traits ³ n'existent pas en Smalltalk originel, la mise en oeuvre présentée ici est celle de *Pharo*.

Un Trait est un ensemble de méthodes héritées par toute classe l'utilisant.

```
Trait named: #TSortable
uses: {}
category: 'Collections-Abstract'

!TSortable methodsFor: 'sorting'!
sort
    "Sort this collection into ascending order using the '<=' operator."
self sort: [:a:b|a<=b]</pre>
```

^{3.} Traits : Composable Units of Behaviour, Nathanael Schärli, Stéphane Ducasse, Oscar Nierstrasz, and Andrew P. Black, In : Cardelli L. (eds) ECOOP 2003 - Object-Oriented Programming. ECOOP 2003. Lecture Notes in Computer Science, vol 2743. Springer.

```
Trait named: #TSortable
uses: {}
category: 'Collections-Abstract'

!TSortable methodsFor: 'sorting'!
sort
    "Sort this collection into ascending order using the '<=' operator."
self sort: [:a:b|a<=b]</pre>
```

- La relation d'utilisation entre une classe et un trait est indépendante de la rélation d'héritage entre classes.
- Une classe peut utiliser plusieurs traits (sous réserve de gérer les conflits de nom).
- Un trait peut utiliser un autre trait.
- Un trait peut être utilisé par un(e) ou plusieurs classes ou traits.
- Une méthode définie sur un trait n'a pas accès à la partie privée des objets.

```
Collection subclass: #SequenceableCollection
uses: TSortable
instanceVariableNames: ''
classVariableNames: ''
package: 'Collections-Abstract'
```

Listing (6) – En utilisant le trait TSortable, la classe SequenceableCollection hérite la méthode sort, toutes ses instances peuvent l'utiliser.

```
c := OrderedCollection new.
d cadd: 1; add: 19; add: 8; add: 11.

c sort -> an OrderedCollection(1 8 11 19)

c isKindOf: TSortable -> false
```

Listing (7) — Utilisation de la méthode sort du trait TSortable pour une instance de OrderedCollection, sous-classe de SequenceableCollection.

3.12 Traits avec état

A suivre ...

4 Le tout-objet et la méta-programmation

4.1 Définitions

Réflexivité : Capacité qu'a un système à donner à ses utilisateurs une représentation de lui-même en connexion causale avec sa représentation effective en machine

méta-entité : entité constitutive du méta-niveau donc du langage ou de son environnement

Entité de première classe : entité, possiblement une méta-entité, ayant une représentation accessible dans un programme, que l'on peut référencer, manipuler et inspecter, et éventuellement modifier.

En Smalltalk, les classes, les méthodes compilées, les méthodes, le source des méthodes, certaines structures de contrôle, éventuellement la pile d'exécution, l'environnement de programmation, le compilateur, ... sont des entités de première classe.

méta-programmation : programmation des entités constitutives du système (méta-entités) en utilisant le système.

4.2 Méta-programmation basique

4.2.1 Les classes comme "rvalues" standards

La "généricité paramétrique" est la capacité à contraindre les structures de données composites, comme les collections par exemple, à ne contenir que des éléments appartenant à des types identifiées dans le texte du programme.

En typage statique (voir les génériques Java, les templates C++) la vérification de telles contraintes peut être faite à la compilation.

Les classes comme "rvalues", accompagnées d'une primitive de test de sous-typage, permettent en typage dynamique un contrôle dynamique de types.

Exemple:

```
Pile subclass: #PileTypee
       instanceVariableNames: 'typeElements'
       classVariableNames: ''
3
       category: 'TP1'
4
   push: element
6
       (element isKindOf: typeElements)
7
           ifTrue: [ super push: element ]
8
           ifFalse: [ self error: 'Impossible d''empiler ', element printString ,
9
                                       ' dans une pile de ', typeElements printString
10
```

Listing (8) – test du type d'un objet stocké dans une variable. A ne pas confondre avec instanceOf de Java.style

4.2.2 Manipulation standard des objets primitifs (rock-bottom objects)

Chaque élément d'un type primitif (nombres, caractères, booléens, chaînes, tableaux) est représenté par un méta-objet qui permet son utilisation comme un objet standard. On peut en particulier lui envoyer des messages.

```
5 factorial "entier"
2 'abcde' at: 3 "chaîne de caractère"
3 #($a $b) reverse "tableau"
4 true ifTrue: [#ofCourse] "booléen"
5 false ifFtrue: [#useless]
6 nil isNil "null pointer or UndefinedObject''
7 selecteur := #facto , #rial "concaténation de symboles"
```

Un type primitif est représenté par une classe mais son implantation n'est pas entièrement définie par cette classe et réside partiellement dans la machine virtuelle.

Il est possible de modifier les méthodes de ces classes (forcément dangereux) et d'en créer de nouvelles (exemple, la méthode factorial sur la classe Integer).

La hiérarchie des classes représentant les nombres.

```
Magnitude
       Number ()
2
           FixedPoint ('numerator' 'denominator' 'scale')
3
           Fraction ('numerator' 'denominator')
4
5
            Integer ()
                LargeInteger ()
6
                   LargeNegativeInteger ()
                   LargePositiveInteger ()
8
                SmallInteger ()
9
           LimitedPrecisionReal ()
10
                Double ()
11
                Float ()
12
```

4.2.3 L'objet nil.

 \mathtt{nil} (\mathtt{null} en Java) est la valeur par défaut du type référence, donc affectée à toute l-value (variable, case de tableau, etc) non explicitement initialisée.

Nil est l'unique instance (Singleton) de la classe UndefinedObject sur laquelle il est possible de définir des méthodes.

Il est donc possible d'envoyer un message à nil.

L'exception nullPointerException n'existe pas!

Le nom de la classe est certainement discutable.

```
nil isNil "true"
nil class "UndefinedObject"
```

Application à l'implantation des collections

La classe UndefinedObject offre une alternative originale pour l'implantation de certains types récursifs, par exemple List ou Arbre.

```
type List =
Empty |
Tuple of Object * List
```

dont certaines fonctions se définissent par une répartition sur les différents constructeurs algébriques :

```
length :: List -> Int
length Empty = 0
length Tuple val suite = 1 + (lentgh suite)
```

Mise en oeuvre (extrait 1)

Mise en oeuvre (extrait 2)

```
!UndefinedObject methodsFor: 'ListManipulation'!
addFirst: element
    ^List with: element

length
    ^0

append: aList
    ^aList
```

4.2.4 Définition de nouvelles structures de contrôle - Plus sur les fermetures lexixales

Les structures de contrôle sont réalisées en Smalltalk par des méthodes définies sur les classes Boolean (conditionnelles, et, ou), *BlockClosure* (boucle "tantque"), Integer (boucle "for"), et utilisant des fermetures lexicales.

Une fermeture lexicale est une fonction anonyme créée dans l'environnement lexical de sa définition et exécutée dans cet environnement quel que soit son point d'activation.

Les fermetures en Smalltalk sont implantées en Smalltalk par la classe BlockClosure.

— Une fermeture est une constante littérale dont la valeur est elle-même.

```
[:x | x + 1] = [:x | x + 1]
```

— Une fermeture peut ainsi être utilisée comme lvalue (affectation, passage en argument) :

```
V := [:x \mid x+1]
(date = '24/12') ifTrue: [Transcript show: 'Noël !!']
```

— Une fermeture est une fonction, que l'on peut donc invoquer. Comme c'est par ailleurs aussi un objet (tout est objet) on l'invoque en lui envoyant un message, ici le message value.

```
 [2+3] \text{ value} 
 = 5
```

— Si la fermeture possède des paramètres, on l'invoque en lui passant des arguments avec les messages value: , value:value:, etc (c'est moins beau qu'en *Scheme* mais c'est conforme au paradigme).

```
V value: 23
=24
[:i | i + 1] value: 33
= 34
```

4.2.5 Fermetures et retour de méthode

Une fermeture capture le point de retour du block ou elle est créée.

```
"une méthode de classe définie sur la classe Object
    qui créée une fermeture et la passe en argument à la méthode test2"
2
   test2Main
3
        self test2: [^1 + 2].
4
        Transcript show: '33333333'; cr.
5
    "la méthode de classe test2 définie également sur Object
7
    invoque la fermeture"
8
   test2: aBlock
9
        Transcript show: '111111111'; cr.
10
        aBlock value.
11
        Transcript show: '22222222'; cr.
12
    "exécution de Object test2Main"
14
   Object test2Main
15
   11111111
16
   =3
17
```

A condition que le point de retour existe toujours lors de l'exécution du block.

```
test1Main
    | b |
    b := self test1.
    Transcript show: '111111111'; cr.
    Transcript show: b value.
    Transcript show: '222222222'; cr.

test1
    ^[^1 + 2]
```

— The receiver tried to return result to homeContext that no longer exists.

4.2.6 Fermeture et Structures de contrôles

De par leur auto-évaluation, les fermetures lexicales permettent la définition de structures de contrôle.

```
"une méthode abstraite sur la classe Boolean"

ifTrue: alternativeBlock

"If the receiver is false (i.e., the condition is false), then the value is the false alternative, which is nil. Otherwise answer the result of evaluating the argument, alternativeBlock. Create an error notification if the receiver is nonBoolean. Execution does not actually reach here because the expression is compiled in—line."

self subclassResponsibility
```

4.3 Les Méta-classes en Smalltalk

4.3.1 Rappel: la solution Objvlisp




Figure (7) – Une classe et une méta-classe sont structurellement et fonctionnellement identiques

- Object est la racine de l'arbre d'héritage,
- Class est la racine de l'arbre d'instantiation, elle est instance d'elle-même et sous-classe de Object.

On parle à son propos d'un système à méta-classes explicites car le créateur d'une classe peut choisir la méta-classe qu'il souhaite instancier.

L'intérêt de cette solution est de permettre le découplage entre la superclasse de la méta-classe et la méta-classe de la super-classe. Une classe peut ainsi et par exemple être abstraite (instance de la méta-classe AbstractClass) sans que ses sous-classes le soient.

L'inconvénient de cette solution est qu'elle peut poser des problèmes de compatibilité (voir section 4.3.8). Le langage *CLOS* qui réalise un sur-ensemble de la solution *Objvlisp* laisse au développeur le soin de résoudre ces problèmes éventuels de compatibilité via la définition de fonctions ad.hoc.

4.3.2 Motivations de la solution Smalltalk

La solution Smalltalk a été conçue avec deux objectifs ⁴:

- Donner la possibilité de définir des méthodes sur les métaclasses tout en rendant les métaclasses invisibles au développeur qui n'a pas envie de les voir. L'invisibilité des méta-classes pour le développeur standard est réalisé par l'éditeur de programmes également nommé browser.
- Rendre impossible la création de métaclasses incompatibles.

^{4.} Il y a donc deux façons de considérer ce système, la première est de l'utiliser sans chercher à entrer dans le détail de sa réalisation et d'apprécier son efficacité. Le bouton "class" du browser est dédié à la programmation des méthodes de classe et il n'est pas nécessaire de comprendre la mécanique interne pour l'utiliser. La seconde est d'aller y voir de plus près pour le plaisir de comprendre, ce qui prend tout son sens dans le cadre d'un cours sur la réflexivité.

4.3.3 Architecture du système de métaclasses




Figure (8) – Classes et Métaclasses : Hiérarchies d'héritage et d'instantiation. (figure : Gabriel Pavillet)




Figure (9) - Classes et Métaclasses : couches d'instantiation. (figure : Gabriel Pavillet)

Les points clé:

- Toute classe C possède une unique métaclasse référençable via l'expression C class.
- Les deux hiérarchies d'héritage entre classes et d'héritage entre métaclasses sont isomorphes.
- Object class est la racine de la sous-hiérarchie des métaclasses.
- Le graphe d'héritage étant un arbre avec donc une racine unique (Object), le lien entre les hiérarchies de classes et de métaclasses est fait au niveau de la métaclasse Object class qui hérite de la classe Class.

Ceci se comprend, selon la sémantique usuelle ⁵ du lien d'héritage, ainsi : une Object class, par exemple Object), est une sorte de classe.

— Il y a 4 niveaux (conceptuels) d'instantiation. Un objet (couche 1) (par exemple un point), instances d'une classe (dans cet exemple Point), instance de l'unique méta-méta-classe (Metaclass).

Le problème de regression infinie dans les descriptions est reglé par le fait que metaclass est instance de metaclass class, qui est une métaclasse standard de la couche 3, instance de Metaclass. Metaclass class a exactement le même format que toutes les autres métaclasses.

4.3.4 Structure des classes et des métaclasses (cf. fig. 10)

```
ProtoObject #()

Object #()

Behavior #('superclass' 'methodDict' 'format')

ClassDescription #('instanceVariables' 'organization')

Class #('subclasses' 'name' 'classPool' 'sharedPools' 'environment' 'category')

[ ... all the Metaclasses ... ]

Metaclass #('thisClass')
```

Figure (10) – Les classes "Class" et "Metaclass" (et leurs attributs) définissant les classes et les méta-classes. Elles ont en commun tout ce qui est hérité de "ClassDescription".

La classe Class déclare, en fermant transitivement la relation "est-sous-classe-de", les attributs : "superclass methodDict format instanceVariables organization subclasses name classPool environment category sharedPools". Toute sorte de Class, pourra posséder une valeur pour chacun d'eux.

Par exemple la valeur de l'attribut instanceVariables de la classe Point (instance de Point class qui hérite de Class) est 'x y z'. Les variables de classes sont rangées dans l'attribut classPool.

La classe Metaclass déclare les attributs : "superclass methodDict format instanceVariables organization this-Class".

thisClass (nom peu évocateur) référence la classe, unique instance (schéma Singleton)) d'une métaclasse dans un contexte donné.

this Class est utilisable dans les méthodes d'instance de la classe Metaclass qui n'a aucune sous-classe ⁶.

^{5.} AKO : a kind of, est une sorte de ... une voiture est une sorte de véhicule.

^{6.} Exercice : créer une sous-classe MC de la classe Metaclass, qui définirait un nouveau type généraliste de meta-classes. La difficulté principale n'est pas dans la création de cette classe mais, comme les metaclasses sont crées automatiquement par le système, de modifier le système de création des classes pour qu'une classe qui en relève ne soit pas une instance de Metaclass mais de MC.




Figure (11) – l'Inspection d'une classe montre que ses champs sont conformes à la déclaration des attributs réalisés dans la classe Class.

4.3.5 Variable de classe et variable d'instance de métaclasse

L'analyse de la structure des classe montre la différence entre variable de classe et variable d'instance ⁷ de métaclasse. A noter qu'il n'y a par ailleurs pas la même dichotomie entre méthode de classe et méthode d'instance de métaclasse.

Les méthodes de classe d'une classe C sont les méthodes d'instance de C class. Les variables de classes d'une classe C n'ont rien à voir avec les variables d'instance de C class.

Les variables de classe sont accessibles dans les méthodes d'instance et dans les méthodes de classe. Elle servent à stocker des valeurs communes à toutes les instances d'une même classe, sans avoir à passer par la métaclasse.

Les variables d'instance de métaclasse ne sont accessibles que dans les méthodes de classe. Ce sont les attributs définis par les métaclasses.

4.3.6 Utilisation standard des méta-classes : 1) création d'instances initialisées

Une classe complète avec ses méthodes de création d'instances : la classe Pile.

```
Object subclass: #Pile
instanceVariableNames: 'index buffer'
classVariableNames: ''
poolDictionaries: ''
category: 'Exercices'
...
```

^{7.} On rappelle que le terme "variable d'instance" dénote ce qui est usuellement nommé "attribut" dans les autres langages.

```
1
    Pile class
2
            instanceVariableNames: ''
3
    !Pile class methodsFor: 'creation'!
5
6
    example
            "self example"
7
            "Rend une pile d'entiers de taille 10, pleine."
8
            aStack
9
            aStack := self new: 10.
10
            1 to: aStack size do: [:i | aStack push: i].
11
12
    new
14
            "On fixe la taille par défaut à 10"
15
            *super new initialize: 10
16
    new: taille
            super new initialize: taille
19
    taille: taille
21
            ^self new: taille
22
```

4.3.7 Utilisation standard des méta-classes : 2) création de classes ayant des attributs additionnels

Si l'on souhaite faire d'une classe une "memo classe" afin qu'elle mémorise la liste de ses instances, la solution naturelle est d'utiliser une variable d'instance de métaclasse.

```
EtudiantMemo class
        instanceVariableNames: 'tous'
2
   initialize
   tous := OrderedCollection new.
    new
       | i |
8
       i := super new.
9
       tous add: i.
10
        î
11
13
    tous
       ^tous
14
```

4.3.8 Avantage du modèle Smalltalk : la compatibilité des métaclasses

La solution Smalltalk ne pose pas de problèmes de compatibilité.

Les problèmes de compatibilité surviennent avec les architectures à la Objvlisp lors de phase d'abstraction (passage au niveau meta - compatibilité ascendante) ou de la phase de réification (passage au niveau réifié - compatibilité descendante).

Problème de compatibilité Ascendante

Soient:

- une classe A, définissant une methode foo : 'self class bar'
- une classe B sous-classe de A
- la classe de A et sa méthode bar : '...'
- la classe de B, non sous-classe de la classe de A, et ne définissant pas de méthode bar
- une instance b de B

alors : 'b foo' lève une l'exception "B ne comprend pas le message bar".

Cette situation est impossible avec le modèle *Smalltalk* où la classe de B ne peut pas ne pas être une sous-classe de la classe de A et hérite donc de la méthode bar.

Problème de Compatibilité Descendante

Soient:

- une méta-classe MA définissant une méthode bar : 'self new foo'
- une méta-classe MB, sous classe de MA
- une classe B instance de MB

'MB bar' peut lever une exception.

4.3.9 Inconvénient du modèle Smalltalk

Il est impossible de découpler les hiérarchies de classes et de méta-classes. Ce qui pose des problèmes d'expression.

Par exemple si on crée une méta-classe AbstractClass dont les instances ne peuvent être instanciées (par une redéfinition appropriée de new). Alors toutes les sous-classes d'une instance de AbstractClass hériteront de cette redéfinition de new. Or, les classes abstraites ont bien sûr des sous-classes concrètes.

4.4 Les Méta-objets permettant d'accéder à la pile d'exécution

Smalltalk permet de manipuler chaque bloc (frame) de la pile d'exécution comme un objet de première classe avec une politique de "si-besoin" car la réification est une opération coûteuse.

Par exemple, le code suivant implante les structures de contrôle catch et throw permettant de réaliser des échappements à la Lisp (réalisation de branchements non locaux), que l'on peut voir comme les structures de contrôle de base nécessaires à l'implantation de mécanismes de gestion des exceptions.

catch permet de définir un point de reprise à n'importe quel point d'un programme et returnToCatchWith: interromp l'exécution strandard et la fait reprendre à l'instruction qui suit le catch correspondant (le symbole receveur détermine la correspondance).

Exemple d'application : interruption d'une recherche dés que l'on a trouvé l'élément recherché pour revenir à un point antérieur de l'exécution du programme.

4.4.1 Version Visualworks7

```
!Symbol methodsFor: 'catch-throw'!
    catch: aBlock
3
        "execute aBlock with a throw possibility"
       aBlock value.
   returnToCatchWith: aValue
           "Look down the stack for a catch, the mark of which is self,
8
           when found, transfer control (non local branch)."
           catchMethod currentContext
10
           currentContext := thisContext.
11
           catchMethod := Symbol compiledMethodAt: #catch:.
12
           [currentContext method == catchMethod and: [currentContext receiver == self]]
13
                   whileFalse: [currentContext := currentContext sender].
14
           thisContext sender: currentContext sender.
15
            aValue
16
```

Listing (9) – version Visualworks-Smalltalk

4.4.2 Version Pharo6

```
!Symbol methodsFor: 'catch-throw'!
   catch: aBlock
       "execute aBlock with a throw possibility"
       aBlock value.
5
   returnToCatchWith: aValue
7
           catchMethod currentContext
8
9
           currentContext := thisContext.
           catchMethod := Symbol compiledMethodAt: #catch:.
10
           [currentContext method == catchMethod and: [currentContext receiver == self]]
11
                  whileFalse: [currentContext := currentContext sender].
12
           currentContext return: aValue.
13
            aValue
14
```

Listing (10) - version Pharo-6

Exemple d'utilisation:

```
#Essai catch: [
Transcript show: 'a';cr.
Transcript show: 'b';cr.
Transcript show: 'c';cr.
#Essai returnToCatchWith: 22.
Transcript show: 'd';cr.
33]
```

Listing (11) – should display a b c (not d) in the Transcript and return 22 (not 33)

4.5 Méta-objets pour accéder au compilateur et aux méthodes compilées

Le compilateur Smalltalk est écrit en Smalltalk et est un objet de première classe. Cela permet de définir dynamiquement de nouvelles classes et méthodes.

L'exemple suivant est extrait de la réalisation d'un mini-tableur en Smalltalk. Dans cet exemple, les formules associées aux cellules sont représentées par des méthodes définies dynamiquement par le programme, lexicalement analysées et compilées à la volée.

Pour savoir comment ajouter, une fois compilée, une méthode à une classe, il faut étudier les protocoles définis sur les classes Behavior et ClassDescription.

```
Model subclass: #Cellule
instanceVariableNames: 'value formula internalFormula dependsFrom '
classVariableNames: ''
poolDictionaries: ''
category: 'Tableur'!
```

```
!Cellule methodsFor: 'compile formula'!
1
   compileFormula: s
2
       "Analyse lexicale, puis syntaxique puis generation de code pour la formule s"
3
       tokens newDep interne methodNode
4
       tokens := Scanner new scanTokens: s.
5
       newDep := (tokens select: [:i | self isCaseReference: i]) asSet.
6
       7
       newDep do: [:each | interne := interne , each , ' '].
8
       interne := interne , '|\ '.
9
       newDep do: [:each | interne := interne , each , ' := (Tableur current at: #' , each , ') value.\ '].
10
       interne := (interne , ' , s) withCRs asText.
11
       methodNode := UndefinedObject compilerClass new
12
                  compile: interne
13
                  in: UndefinedObject
14
                  notifying: nil
15
                  ifFail: [].
16
       internalFormula := methodNode generate.
17
       ^newDep
18
```

Remerciements

Merci à René Paul Mages pour ses relectures attentives et remarques pertinentes. https://renemages.wordpress.com/smalltalk/pharo/